

Chapitre V

Fonctions

1 Définition des fonctions

1.1 Fonctions prédéfinies

Les fonctions usuelles sont prédéfinies. Ainsi, on peut utiliser `sin`, `exp`, `ln` etc. Nous avons déjà utilisé ces fonctions.

1.2 Opérateur flèche (->)

Pour définir une fonction, la première méthode consiste à utiliser l'opérateur flèche (noté par moins-strictement supérieur). La syntaxe est la suivante :

$$\text{\texttt{`nom de fonction` :=var(s) ->expression en var(s)}}$$

Effectuer :

```
>restart
>f:=x->2*x^2+3;
>f(5);
>f(a+b);
>expand(%);
>g:=(x,y)->cos(x-y^2);
>g(Pi,sqrt(3*Pi)/2);
```

1.3 Fonctions définies par morceaux

On peut définir une fonction par morceaux à l'aide de l'opérateur flèche (->) et de la fonction `piecewise`. La syntaxe pour définir :

$$f : x \mapsto \begin{cases} \textit{expr1} & \textit{si cond1} \\ \textit{expr2} & \textit{si cond2} \\ \vdots & \\ \textit{exprn} & \textit{sinon} \end{cases}$$

est :

$$\text{\texttt{f:=x->piecewise(cond1,expr1,cond2,expr2, ...,exprn)}}$$

Effectuer :

```
>f:=x->piecewise(x<0,-x^2+1,x<1,-x+1,x^2-1);
>plot(f,-2..2);
```

1.4 Utilisation de `unapply`

C'est la fonction à utiliser pour transformer une expression en fonction.

$$\text{\texttt{f:=unapply(expression,variable)}}$$

Effectuer :

```
>expr:=3*x+4;
>f:=x->expr; # C'est un premier essai
>f(2);
>f:=unapply(expr,x); # C'est la bonne façon de faire
>f(2);
```

Remarque. L'expression qui suit l'opérateur flèche n'est pas évaluée. Cela justifie les résultats qui suivent et qui montre l'un des intérêts de la fonction `unapply`.

Effectuer:

```
>restart;
>L:= [seq(x->x^k,k=1..4)];
>L[2]; # Ce n'est pas ce qui était attendu
>L1:= [seq(subs(n=k,x->x^n),k=1..4)]; # 1ère méthode qui marche
>L2:= [seq(unapply(x^k,x),k=1..4)]; # 2nde méthode qui marche
>L1[3](t);L2[3](t);
```

1.5 Utilisation des procédures

On étudiera plus en détail les procédures. Les fonctions, définies comme ci-dessus, sont de premiers exemples de procédures. L'utilisation des procédures permet de définir des fonctions d'une façon plus compliquée.

Effectuer:

```
>signe:=proc(x::realcons)
if x>0 then "+" elif x<0 then "-" else "0" fi
end;
>signe(3);signe(-1);
```

2 Opérations sur les fonctions

2.1 Composition

On peut composer deux fonctions à l'aide de l'opérateur `@`. On peut composer une fonction n -fois avec elle-même en utilisant `@@n`.

Effectuer:

```
>g:=x->x+Pi/2;
>f:=cos@g;
>simplify(f(x));
>h:=sin@(g@@4);
>simplify(h(x));
```

2.2 Tracés de représentations graphiques

2.2.1 À partir d'une fonction

On utilise la fonction `plot` qui s'utilise en respectant la syntaxe suivante :

`plot(fonction , a..b , c..d , options)`

Les trois derniers arguments sont optionnels. Par défaut, l'intervalle horizontal `a..b` est fixé à `-10..10`, tandis que l'intervalle vertical `c..d` est calculé par Maple.

Attention ! Il faut que la fonction ne dépende d'aucun paramètre. Si la fonction que l'on veut représenter dépend d'un paramètre, on substitue à ce paramètre une valeur numérique à l'aide de `subs`.

Effectuer:

```
>plot(sin);
>plot(x->x^2*sin(1/x),-1..1);
```

```
>plot(t->3*a*ln(t),0..1);
>H:=omega->1/(1+(I*omega)/(Q*omega0)-omega^2/omega0^2);
# avec  $Q = \frac{1}{R}\sqrt{\frac{C}{L}}$  et  $\omega_0 = \frac{1}{\sqrt{LC}}$ 
>Hnum:=unapply(subs(Q=1,omega0=1000,H(omega)),omega);
>plot(abs@Hnum,0..3000);
>with(plots);
>semilogplot(20*log[10]@abs@Hnum,1..10^6,title="Diagramme de Bode");
```

2.2.2 À partir d'une expression

On utilise encore la fonction `plot`, mais en respectant la syntaxe suivante :

```
plot(expression_en_x , x=a..b , c..d , options)
```

Effectuer :

```
>plot(cos(x),x);
>P:=a*x^2+b*x+c;
>Pnum:=subs(a=1,b=-2,c=4,P);
>plot(Pnum,x=-5..5);
```

2.2.3 Les options de plot

On peut explorer les options suivantes (se référer à l'aide)

1. `discont=true` ou `discont=false` (défaut : `false`)
2. `numpoints=100` (défaut : 50)
3. `resolution=100` (défaut : 200)
4. `color=green` (défaut : `red`)
5. `title="Mon beau graphique"` (défaut : `""`)
6. `scaling=constrained`
7. On peut changer l'aspect du graphe après l'avoir tracé, en utilisant les boutons de la barre d'outils.

Tester les différentes options sur la représentation graphique de tangente.

2.2.4 Représentation simultanée de plusieurs graphes

Première méthode : Emploi d'un ensemble (ou d'une liste).

Effectuer :

```
>plot({cos(x),1-x^2/2},x=-Pi..Pi);
>approx:=seq(sum(x^k/k!,k=0..n),n=1..5);
>plot([exp(x),approx],x=-3..3,color=[red,blue $ 5]);
```

Deuxième méthode : Utilisation de `display`.

Chaque graphique est rangé dans une variable. La fonction `display` permet alors de présenter tous les graphiques en une seule fois :

```
display([graphe1,graphe2,...,graphen],options)
```

après avoir chargé le module `plots` et défini les variables `graphe1` etc.

Effectuer :

```
>restart;with(plots):
>g1:=plot(cos(x),x=-2*Pi..2*Pi,color=magenta):
>g2:=plot(sin(x),x=-2*Pi..2*Pi,color=blue):
```

```
>g3:=plot(tan(x),x=-2*Pi..2*Pi,-3..3,color=green,discont=true):
>display([g1,g2,g3],title="Principales fonctions trigonométriques");
```

2.2.5 Remarque : Tracés de lignes brisées

Effectuer :

```
>plot([[0,0],[1,3],[4,3],[0,0]]);
>plot([[0,0],[2,2],[0,2],[1,3],[2,2],[2,0],[0,2],[0,0],[2,0]],scaling=constrained,
      axes=none,color=aquamarine);
```

2.3 Utilisation de map

La fonction `map` permet d'appliquer une fonction à tous les opérandes d'une expression.

Effectuer :

```
>restart;
>f:=t->t^2
>map(f,a*x+b);
>map(f,t*(a+b));
>map(f,[seq(y||i,i=1..5)]);
```

3 Exercices

V.1 Dans \mathbb{C} , résoudre l'équation $z^2 + \bar{z} + iz = 0$. Montrer que les points dont les affixes sont solutions forment un triangle rectangle que l'on représentera.

V.2 Déterminer les nombres complexes z tels que $\frac{z^2}{2z+3i}$ soit imaginaire pur. Les représenter dans le plan complexe.

V.3

1. Montrer que :

$$f : z \mapsto \frac{z - 2i}{z + i}$$

est une bijection de $\mathbb{C} \setminus \{-i\}$ dans $\mathbb{C} \setminus \{1\}$.

2. On pose $z = x + iy$ et $f(z) = u + iv$, où $(x, y, u, v) \in \mathbb{R}^4$.

Exprimer x et y en fonction de u et v , et inversement.

3. Soit :

$$P = \left\{ z \in \mathbb{C} \text{ t.q. } \operatorname{Im}(z) > \frac{1}{2} \right\} \quad D = \{z \in \mathbb{C} \text{ t.q. } |z| < 1\}$$

Démontrer que f induit une bijection de P sur D .

V.4 Soit f la fonction de \mathbb{C}^* dans lui-même qui à z associe :

$$f(z) = \frac{1}{2} \left(z + \frac{1}{z} \right)$$

1. Déterminer et représenter l'image par f du cercle de centre O et de rayon R .
2. Déterminer et représenter l'image par f de la demi-droite d'origine O et d'angle polaire θ .